// File: contracts/periphery/PreBluejayToken.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/cryptography/MerkleProof.sol";

import "../interfaces/IMintableBurnableERC20.sol";
import "../interfaces/ITreasury.sol";
import "../interfaces/IPreBluejayToken.sol";

import "./MerkleDistributor.sol";

/// @title PreBluejayToken
/// @author Bluejay Core Team
/// @notice PreBluejayToken is the contract for the pBLU token. The token is a
/// non-transferable token that can be redeemed for underlying BLU tokens. The
/// redemption ratio is proportional to the current total supply of the BLU token
/// against the targeted total supply of BLU tokens. The contract allows a mimimal
/// level of tokens to be redeemed by each account for initial liquidity.
/// ie If the target supply is 50M and the current supply is 10M, users will be
/// able to redeem 1/5 of their pBLU tokens as BLU tokens.
/// @dev The pBLU token is not an ERC20 token
contract PreBluejayToken is Ownable, MerkleDistributor, IPreBluejayToken {
  uint256 constant WAD = 10**18;

  /// @notice The contract address of the treasury, for minting BLU
  ITreasury public immutable treasury;

  /// @notice The contract address of the BLU token
  IMintableBurnableERC20 public immutable BLU;

  /// @notice Target BLU total supply when all pBLU are vested, in WAD
  uint256 public immutable bluSupplyTarget;

  /// @notice Amount claimable that does not require vesting, in WAD
  uint256 public immutable vestingThreshold;

  /// @notice Flag to pause contract
  bool public paused;

  /// @notice Mapping of addresses to allocated pBLU, in WAD
  mapping(address => uint256) public quota;

  /// @notice Mapping of addresses to redeemed pBLU, in WAD
  mapping(address => uint256) public redeemed;

  /// @notice Constructor to initialize the contract
  /// @param _BLU Address of the BLU token
  /// @param _treasury Address of the treasury
  /// @param _merkleRoot Merkle root of the distribution
  /// @param _bluSupplyTarget Target BLU total supply when all pBLU are vested, in WAD
  /// @param _vestingThreshold Amount claimable that does not require vesting, in WAD
  constructor(
    address _BLU,
    address _treasury,
    bytes32 _merkleRoot,
    uint256 _bluSupplyTarget,
    uint256 _vestingThreshold
  ) {
    BLU = IMintableBurnableERC20(_BLU);
    treasury = ITreasury(_treasury);
    _setMerkleRoot(_merkleRoot);
    bluSupplyTarget = _bluSupplyTarget;
    vestingThreshold = _vestingThreshold;
    paused = true;
  }

  // =============================== PUBLIC FUNCTIONS =================================

  /// @notice Claims pBLU tokens
  /// @dev The parameters of the function should come from the merkle distribution file.
  /// @param index Index of the distribution
  /// @param account Account where the distribution is credited to
  /// @param amount Amount of pBLU allocated in the distribution, in WAD
  /// @param merkleProof Array of bytes32s representing the merkle proof of the distribution
  function claimQuota(
    uint256 index,
    address account,
    uint256 amount,
    bytes32[] calldata merkleProof
  ) public override {
    _claim(index, account, amount, merkleProof);
    quota[account] += amount;
  }

  /// @notice Redeem BLU token against pBLU tokens
  /// @dev During redemption, the quota does not change. Instead the redeemed amount is
  /// updated to reflect amount of pBLU redeemed.
  /// @param amount Amount of BLU tokens to redeem
  /// @param recipient Address where the BLU tokens will be sent to
  function redeem(uint256 amount, address recipient) public override {
    require(!paused, "Redemption paused");
    require(
      redeemableTokens(msg.sender) >= amount,
      "Insufficient redeemable balance"
    );
    redeemed[msg.sender] += amount;
    treasury.mint(recipient, amount);
    emit Redeemed(msg.sender, recipient, amount);
  }

  // =============================== VIEW FUNCTIONS =================================

  /// @notice Gets the overall vesting progress
  /// @return progress The vesting progress, in WAD
  function vestingProgress() public view override returns (uint256) {
    uint256 bluSupply = BLU.totalSupply();
    return
      bluSupply < bluSupplyTarget ? (bluSupply * WAD) / bluSupplyTarget : WAD;
  }

  /// @notice Gets the amount of BLU that can be redeemed for a given address
  /// @param account Address to get the redeemable balance for
  /// @return redeemableAmount Amount of BLU that can be redeemed, in WAD
  function redeemableTokens(address account)
    public
    view
    override
    returns (uint256)
  {
    uint256 quotaVested = (quota[account] * vestingProgress()) / WAD;
    if (quotaVested <= vestingThreshold) {
      quotaVested = quota[account] < vestingThreshold
        ? quota[account]
        : vestingThreshold;
    }
    return quotaVested - redeemed[account];
  }

  // =============================== ADMIN FUNCTIONS =================================

  /// @notice Pause and unpause the contract
  /// @param _paused True to pause, false to unpause
  function setPause(bool _paused) public onlyOwner {
    paused = _paused;
  }

  /// @notice Set the merkle root for the distribution
  /// @dev Setting the merkle root after distribution has begun may result in unintended consequences
  /// @param _merkleRoot New merkle root of the distribution
  function setMerkleRoot(bytes32 _merkleRoot) public onlyOwner {
    _setMerkleRoot(_merkleRoot);
    emit UpdatedMerkleRoot(_merkleRoot);
  }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/cryptography/MerkleProof.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/cryptography/MerkleProof.sol)

pragma solidity ^0.8.0;

/**
 * @dev These functions deal with verification of Merkle Trees proofs.
 *
 * The proofs can be generated using the JavaScript library
 * https://github.com/miguelmota/merkletreejs[merkletreejs].
 * Note: the hashing algorithm should be keccak256 and pair sorting should be enabled.
 *
 * See `test/utils/cryptography/MerkleProof.test.js` for some examples.
 */
library MerkleProof {
    /**
     * @dev Returns true if a `leaf` can be proved to be a part of a Merkle tree
     * defined by `root`. For this, a `proof` must be provided, containing
     * sibling hashes on the branch from the leaf to the root of the tree. Each
     * pair of leaves and each pair of pre-images are assumed to be sorted.
     */
    function verify(
        bytes32[] memory proof,
        bytes32 root,
        bytes32 leaf
    ) internal pure returns (bool) {
        return processProof(proof, leaf) == root;
    }

    /**
     * @dev Returns the rebuilt hash obtained by traversing a Merklee tree up
     * from `leaf` using `proof`. A `proof` is valid if and only if the rebuilt
     * hash matches the root of the tree. When processing the proof, the pairs
     * of leafs & pre-images are assumed to be sorted.
     *
     * _Available since v4.4._
     */
    function processProof(bytes32[] memory proof, bytes32 leaf) internal pure returns (bytes32) {
        bytes32 computedHash = leaf;
        for (uint256 i = 0; i < proof.length; i++) {
            bytes32 proofElement = proof[i];
            if (computedHash <= proofElement) {
                // Hash(current computed hash + current element of the proof)
                computedHash = keccak256(abi.encodePacked(computedHash, proofElement));
            } else {
                // Hash(current element of the proof + current computed hash)
                computedHash = keccak256(abi.encodePacked(proofElement, computedHash));
            }
        }
        return computedHash;
    }
}

// File: contracts/interfaces/IMintableBurnableERC20.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/interfaces/IERC20.sol";

interface IMintableBurnableERC20 is IERC20 {
  function mint(address _to, uint256 _amount) external;

  function burn(uint256 amount) external;
}

// File: contracts/interfaces/ITreasury.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

interface ITreasury {
  function mint(address to, uint256 amount) external;

  function withdraw(
    address token,
    address to,
    uint256 amount
  ) external;

  function increaseMintLimit(address minter, uint256 amount) external;

  function decreaseMintLimit(address minter, uint256 amount) external;

  function increaseWithdrawalLimit(
    address asset,
    address spender,
    uint256 amount
  ) external;

  function decreaseWithdrawalLimit(
    address asset,
    address spender,
    uint256 amount
  ) external;

  event Mint(address indexed to, uint256 amount);
  event Withdraw(address indexed token, address indexed to, uint256 amount);
  event MintLimitUpdate(address indexed minter, uint256 amount);
  event WithdrawLimitUpdate(
    address indexed token,
    address indexed minter,
    uint256 amount
  );
}

// File: contracts/interfaces/IPreBluejayToken.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

interface IPreBluejayToken {
  event Redeemed(
    address indexed owner,
    address indexed recipient,
    uint256 amount
  );

  event UpdatedMerkleRoot(bytes32 merkleRoot);

  function claimQuota(
    uint256 index,
    address account,
    uint256 amount,
    bytes32[] calldata merkleProof
  ) external;

  function redeem(uint256 amount, address recipient) external;

  function redeemableTokens(address account) external view returns (uint256);

  function vestingProgress() external view returns (uint256);
}

// File: contracts/periphery/MerkleDistributor.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.4;

import "@openzeppelin/contracts/utils/cryptography/MerkleProof.sol";

/// @title MerkleDistributor
/// @author Bluejay Core Team
/// @notice MerkleDistributor is a base contract for contracts using merkle tree to distribute assets.
/// @dev Code inspired by https://github.com/Uniswap/merkle-distributor/blob/master/contracts/MerkleDistributor.sol
/// Merkle root generation script inspired by https://github.com/Uniswap/merkle-distributor/tree/master/scripts
abstract contract MerkleDistributor {
  /// @notice Merkle root of the entire distribution
  /// @dev Setting the merkle root after distribution has begun may result in unintended consequences
  bytes32 public merkleRoot;

  /// @notice Packed array of booleans
  mapping(uint256 => uint256) private claimedBitMap;

  event Distributed(uint256 index, address account, uint256 amount);

  /// @notice Checks `claimedBitMap` to see if the distribution to a given index has been claimed
  /// @param index Index of the distribution to check
  /// @return claimed True if the distribution has been claimed, false otherwise
  function isClaimed(uint256 index) public view returns (bool) {
    uint256 claimedWordIndex = index / 256;
    uint256 claimedBitIndex = index % 256;
    uint256 claimedWord = claimedBitMap[claimedWordIndex];
    uint256 mask = (1 << claimedBitIndex);
    return claimedWord & mask == mask;
  }

  /// @notice Internal function to set a distribution as claimed
  /// @param index Index of the distribution to mark as claimed
  function _setClaimed(uint256 index) private {
    uint256 claimedWordIndex = index / 256;
    uint256 claimedBitIndex = index % 256;
    claimedBitMap[claimedWordIndex] =
      claimedBitMap[claimedWordIndex] |
      (1 << claimedBitIndex);
  }

  /// @notice Internal function to claim a distribution
  /// @param index Index of the distribution to claim
  /// @param account Address of the account to claim the distribution
  /// @param amount Amount of the distribution to claim
  /// @param merkleProof Array of bytes32s representing the merkle proof of the distribution
  function _claim(
    uint256 index,
    address account,
    uint256 amount,
    bytes32[] calldata merkleProof
  ) internal {
    require(!isClaimed(index), "Already claimed");

    // Verify the merkle proof.
    bytes32 node = keccak256(abi.encodePacked(index, account, amount));
    require(MerkleProof.verify(merkleProof, merkleRoot, node), "Invalid proof");

    // Mark it claimed
    _setClaimed(index);

    emit Distributed(index, account, amount);
  }

  /// @notice Internal function to set the merkle root
  /// @dev Setting the merkle root after distribution has begun may result in unintended consequences
  function _setMerkleRoot(bytes32 _merkleRoot) internal {
    merkleRoot = _merkleRoot;
  }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}